

Van der Polin yhtälö


RLC-virtapiirissä oleva vastus vaikuttaa varsin olennaisesti piirissä esiintyviin värähtelyilmiöihin. Kuitenkin aivan uuden elementin komponenttitekniikkaan toivat aikoinaan puolijohdediodeja edeltäneet *tyhjiöputket*. Tyhjiöputken toiminta perustuu siihen, että putken katodia hehkutetaan ohjausjännitteellä, joka saa sen elektronit karkaamaan tyhjiöön. Putken yli olevan jännitteen ollessa myötäsuuntaan, elektronit etenevät anodille ja virta kulkee putken läpi. Korkeilla virran arvoilla rajoitettu kuljettajaelektronijoukko aiheuttaa ohmisen vastuksen putken läpi kulkeville elektroneille; tämä vastus kasvaa virran myötä. Virran ollessa heikko putkessa oleva hehkutusjännite ja sen vapauttavat elektronit helpottavat virran kulkua ja putki toimii negatiivisen vastuksen tavoin. Tämä muodostaa uuden, mielenkiintoisen ilmiön: itseaiheutetut oskillaatiot. Näiden oskillaatioiden kuvaamiseen käytetään *van der Polin* yhtälöä.

Tilanteen lähtökohta on hyvin samankaltainen kuin RLC-virtapiirin yhtälön muodostamisessa. Piirissä syntyvät jännitehäviöt perustuvat induktanssiin, resistanssiin ja kapasitanssiin. Kun yhtälö muodostetaan, nämä tekijät ovat

$$\begin{aligned} E_L &= L \frac{d I(t)}{dt} && \text{käämin yli,} \\ E_R &= \mu \left(\frac{1}{3} I(t)^3 - I(t) \right) && \text{tyhjiöputken yli sekä} \\ E_C &= \frac{1}{C} \int I(t) dt && \text{kondensaattorin yli.} \end{aligned}$$

Ainoa ero syntyy siis tyhjiöputken vastuksesta, joka vakioresistanssin sijaan riippuukin virran suuruudesta yllä esitetyllä tavalla. Virta on vastustermissä merkitty suurettomaksi, siis itse asiassa $\frac{I(t)}{[A]}$. Piirretään kuvaaja tyhjiöputken ohmisen vastuksen virtariippuvuudelle.

`Plot[(i^3)/3 - i, {i, 0, 3}]`



- Graphics -

Vastus on siis alhaisilla virran arvoilla negatiivinen, ja tyhjiöputken hehkutuksen jännitelähde syöttää energiaa systeemiin. Korkeammilla virran arvoilla vastus muuttuu positiiviseksi.

Piirin yhtälö on

$$E(t) = L \frac{d I(t)}{dt} + \mu \left(\frac{1}{3} I(t)^3 - I(t) \right) + \frac{1}{C} \int I(t) dt.$$

Jos piiriin ei syötetä alkuhetken jälkeen ulkoista jännitettä ($E(t)=0$), saadaan virran suhteen derivoimalla homogeeniyhtälö

$$L \frac{d^2 I(t)}{dt^2} + \mu (I(t)^2 - 1) \frac{d I(t)}{dt} + \frac{1}{C} I(t) = 0.$$

Mikäli induktanssi ja kapasitanssi skaalataan ykkösiksi, tämä on perinteinen *van der Polin* yhtälö. Ratkaistaan yhtälön virta ajan funktiona.

Laskujen aluksi on syytä hävittää mahdollisista aiemmista laskuista jääneet muuttujat:

```
Remove["Global`*"]
```

Yhtälö normaaliryhmän muotoon kirjoitettuna on

```
ryhma = {i'[t] == v[t], L v'[t] + μ (i[t]^2 - 1) v[t] + i[t] / C == 0}
{i'[t] == v[t],  $\frac{i[t]}{C} + \mu (-1 + i[t]^2) v[t] + L v'[t] == 0$ }
tuntemattomat = {i[t], v[t]}
{i[t], v[t]}
```

Annetaan vakioille arvot ja muodostetaan alkuehto:


```
ryhma1 = ryhma /. {L -> 0.1, C -> 0.001, μ -> 0.5}
{i'[t] == v[t], 1000. i[t] + 0.5 (-1 + i[t]^2) v[t] + 0.1 v'[t] == 0}
alkuehto1 = {i[0] == 0.5, v[0] == 0.0}
{i[0] == 0.5, v[0] == 0.}
```

Ratkaistaan yhtälöryhmä ja sievennetään tulos:

```
rkt = NDSolve[Flatten[{ryhma1, alkuehto1}], tuntemattomat, {t, 0, 2}, MaxSteps -> 3500]
{{i[t] -> InterpolatingFunction[{{0., 2.}}, <>][t],
  v[t] -> InterpolatingFunction[{{0., 2.}}, <>][t]}}
virta1 = tuntemattomat /. First[rkt]
{InterpolatingFunction[{{0., 2.}}, <>][t], InterpolatingFunction[{{0., 2.}}, <>][t]}
```

Piirretään kuvaaja virralle:

```
Plot[virta1[[1]], {t, 0, 2}, PlotStyle -> RGBColor[0, 0, 1],
GridLines -> Automatic, PlotRange -> {-2.5, 2.5}]
```



- Graphics -

Systeemi käyttäytyy *van der Polin* yhtälöille tyypillisellä tavalla. Se kehittää virran oskillaatiota, joka kasvaa saturoituen raja-arvoonsa. Huomaa kuitenkin, että alkuvirta tai sen muutos on välttämätön vahvistumisen käynnistämiseksi.

Tarkastellaan systeemiä vielä tilanteessa, jossa alkuvirta on saturaatioarvoa merkittävästi suurempi.

```
alkuehto2 = {i[0] == 5.0, v[0] == 0.0}
```

```
{i[0] == 5., v[0] == 0.}
```


Ratkaistaan yhtälöryhmä ja sievennetään tulos.

```
rtk2 = NDSolve[Flatten[{ryhma1, alkuehto2}], tuntemattomat, {t, 0, 1}, MaxSteps -> 3000];
virta2 = tuntemattomat /. First[rtk2]
```

```
{InterpolatingFunction[{{0., 1.}}, <>][t], InterpolatingFunction[{{0., 1.}}, <>][t]}
```

Piirretään virralle kuvaaja.

```
Plot[virta2[[1]], {t, 0, 1}, PlotStyle -> RGBColor[0, 0, 1], GridLines -> Automatic]
```




- Graphics -

Esiin tuli *van der Polin* yhtälön toinen ominaispiirre. Se rajoittaa erittäin tehokkaasti ylisuuria oskillaatioita ja tuo ne saturaatioarvoonsa. Näin tyhjiöputki piiri toimii todella tehokkaana värähtelyn tasaaajana.

Seuraava kuva esittää tilannetta faasitasossa, jolloin akseleilla ovat virta ja sen aikaderivaatta. Tarkastellaan kehitystä matalalla alkuvirralla (vihreä), korkealla alkuvirralla (sininen) sekä tasapainotilanteessa (punainen). Huomaa derivaatan arvon skaalaus tekijällä $\frac{1}{100}$.

```
kuva1 = ParametricPlot[{virta1[[1]], virta1[[2]] / 100},
  {t, 0., 0.8}, DisplayFunction -> Identity, PlotStyle -> RGBColor[0, 1, 0]];
kuva2 = ParametricPlot[{virta2[[1]], virta2[[2]] / 100}, {t, .0, 0.35},
  DisplayFunction -> Identity, PlotStyle -> RGBColor[0, 0, 1]];
raja = ParametricPlot[{virta1[[1]], virta1[[2]] / 100}, {t, 1.9, 2.},
  DisplayFunction -> Identity, PlotStyle -> RGBColor[1, 0, 0]];
Show[kuva1, kuva2, raja, AspectRatio -> Automatic, DisplayFunction -> $DisplayFunction]
```



- Graphics -

Tehtävä

Tämä diagrammi osoittaa hyvin selkeästi *van der Polin* yhtälön käyttäytymistavan. Faasitason tasapainokäyrä on ympyränmuotoinen sopivalla μ :n arvolla, mutta tekijän μ suurentaminen johtaa toisenlaisiin värähtelykuvioihin. Kuinka suureksi μ tulee määrittää, ennen kuin piirin värähtelyssä alkaa esiintyä kulmikkautta? Miksi piiri käyttäytyy siten?

Linkit

[vaihtovirtapiirin vapaa värähtely](#) (virtap2.nb)
[faasitaso](#)

JP & SKK 09.05.2001